



DYNAMIC COLOR NEWS is published monthly by DYNAMIC ELECTRONICS, INC., P.O. Box 896, Hartselle, AL 35640, phone (205) 773-2758. Bill Chapple, President; Alene Chapple, Sec. & Treas.; John Pearson, Ph. D. Consultant; Bob Morgan, Ph. D., Consultant.

Entire Contents (c) by DYNAMIC ELECTRONICS INC., 1984. DYNAMIC COLOR NEWS is intended for the private use of our subscribers and purchasers. All rights reserved. Contents of this newsletter may not be copied in whole or in part without written permission from DYNAMIC ELECTRONICS INC. Subscriptions are \$15/yr for U.S.A. & Canada, \$30 other foreign.

The purpose of this newsletter is to provide instruction on Basic & Machine Language programming, Computer theory, operating techniques, computer expansion, plus provide answers to questions from our subscribers.

The submission of questions, operating hints, and solutions to problems to be published in this newsletter are encouraged. All submissions become the property of Dynamic Electronics if the material is used. We reserve the right to edit all material used and not to use material which we determine is unsuited for publication.

We encourage the submission of Basic and Machine Language Programs as well as articles. All Programs must be well documented so the readers can understand how the program works. We will pay for programs and articles based upon their value to the newsletter. Material sent will not be returned unless return postage is included. Basic & ML programs should be sent on a tape or disk & comments should be sent as a DAT or BIN file.

```
*****
*
*   DYNAMIC   COLOR   NEWS
*
*       December 1984
*
*   Editor and Publisher
*       Bill Chapple
*
*       Secretary
*       Helen Earls
*
*****
```

CONTENTS

Editor's Comments . . . . .	3
Random Numbers . . . . .	4
Computer Generated Sound . . . . .	4
Part 2	
Sound Learning Program . . . . .	6
Sound Demonstration Program . . . . .	7
Sound Program Listing . . . . .	8
Operating Hint . . . . .	9
Next Month . . . . .	9

## EDITOR'S COMMENTS

This month we are continuing our series on Computer Generated Sound. We are including a program that will allow you to hear different sounds. This is fortunate because a lot of times it is not possible to give an example of a concept. With this program you will be able to hear different sounds in your computer and change the time or frequency and duration of them.

The numbers of Color Computer owners should greatly expand due to the lower cost of the computers. With more and more people purchasing them there seems to be a need for more instruction on using them. We had discontinued our Computer School because of a lack of students. However recently several people have contacted me requesting that the school be continued. One of our former students requested to be contacted if we should offer a more advanced course. This is encouraging to us because we can now provide computers for a class without the cash outlay which was required a couple of years ago. For \$100 for a computer and another \$60 to \$70 for a television we can provide equipment for a pair of students. Two years ago we paid \$200 for a 16K computer for our class. So now it is more economical. Television sets are about the same. We bought 13" sets then but I think we will buy 5" sets because the 13" ones took up too much space.

Our classes were conducted similar to the way we do these newsletters. Dr. Pearson helped me for a while and we always showed the students examples of what we were teaching. The

students were very interested and the ages varied from 14 to 70. The 14 year old boy and the 70 year lady did not have any problems. The ones who had the most trouble were people who had not worked in any kind of office environment for a number of years. However I don't remember any student who applied himself and did not learn something.

I think of our readers as being students ready to learn. That is why we write these newsletters with a learning emphasis. Let us know what you want us to cover and we will try to cover what you need. We received a letter from a reader who wants us to give examples of how to write a program that uses bank switching. This kind of information from you is very helpful to us. So we will see if we can write an editorial on this for next month.

The new Color Computers will be harder to upgrade than the older ones. The reason is that Radio Shack had to cut costs to get the price of the computers down. The only chips that have sockets are the memory chips and the SAM chip. For the previous computers about all of the chips had sockets. Troubleshooting will be harder because chips can not be easily removed and replaced without soldering. This also means that modifications that previously involved putting modules under chips will be impossible to do without soldering. However the performance seems to be about the same and they are excellent buys. With the Christmas and New Year's holidays upon us we want to wish each of you a Merry Christmas and a Happy New Year.

## RANDOM NUMBERS

Random numbers are very useful when you want to break up the order of events. When cards are shuffled then their order is confused. For studying notes it is advantages to ask questions in a random order so that the questions and answers can be learned and not the order. Basic has a command for generating a random number. It is as follows:

```
X= RND (Y)
```

X is the random number and Y is the maximum number from which the random number is picked. For example for a set of 52 playing cards if we ordered the cards so that they were numbered from 1 to 52 then we would let Y = 52. If we had 1000 test questions and wanted to randomly pick one then we would let Y = 1000.

## RANDOM SOUNDS

The random number feature of Basic is very useful if you want to create some special sound effects. Sounds can be generated with Color Computers with the following Basic Command:

```
SOUND A, B
```

The first argument "A" is the frequency of the sound and the second argument "B" is the duration. We can randomly pick "A" or "B". Suppose we just want random sounds with a unit length. Then a typical program would be as follows:

```
10 X= RND (255)
20 FOR J = 1 TO 100
30 SOUND X,1
40 NEXT J
```

The preceding program will give 100 random sounds. This is very useful for games or when you want to emphasize something with

random sounds.

## SHUFFLING CARDS

How would you shuffle cards using a computer? There are 4 suits and 13 cards in each suit in a standard deck of cards. The shuffling could be accomplished in two steps. First we could randomly pick a suit. This would be to select a random number out of a maximum possibility of 4. Next we would randomly select a card in the suit. This involves randomly selecting 1 of 13 cards. What if a card has been previously selected? How would you handle it? Another approach would be to number the cards from 1 to 52. Is this easier? We will continue with this discussion next month and give you a program to randomly sort cards. As an exercise see if you can write a program to sort a deck of cards and see how it compares with ours next month.

## COMPUTER GENERATED SOUND - Part 2

Last month we introduced computer generated sound. With computers it is possible to generate just about any combination of voice, music, or noise. An orchestra with many different instruments can be composed and controlled by a computer with appropriate electronic support circuits. We are all familiar with the computer voices we hear from the telephone company when we call information for a number. Also some banks have numbers you can call for the time and/ or weather. This information is given by a computer generated voice. These voices are very good and if you didn't know better you would think that a person were talking to you.

The electronics that allows computers to generate sound are called Digital to Analog (D/A)

converters. Last month we discussed the difference between digital and linear or analog voltages. The output of a D/A converter is a step voltage similar to a staircase. To approximate a linear waveform such as music, a digital word has to be given to the D/A converter at discrete time intervals. Last month we showed how a D/A converter could be used to approximate a sine wave. If we take a large number of approximations in small time intervals then we can get a good approximation from the D/A converter of an analog signal.

There are always tradeoffs that have to be made. If we want very accurate approximations of the vertical or amplitude then we have to have a D/A converter that has a large number of bits.

The D/A converter inside Color Computers has 5 bits which can give 64 discrete levels. This does not give a very good approximation for music or voice. The sound is similar to what you would hear with a talking toy. If the D/A converter were 8 bits then we would have 256 discrete steps. This is enough to give a good approximation for voice and music. D/A converters of 10 bits (1024) steps and 12 bits (4096) steps are available and can give very good approximations of an analog signal. With D/A converters having a high resolution or large number of bits, there is usually a corresponding reduction in speed. So with the graph we drew last month, if we make more vertical steps then we tradeoff horizontal steps which represent time or speed.

#### COCO'S D/A

As stated earlier color computers have a built in D/A converter for generating sound. Let's look at how this converter works and what is needed to

program it.

#### PIA CHIPS

The Peripheral Interface Adapter (PIA) chips are used as input and output buffers for Color Computers. These are 40 pin chips and carry Motorola's part numbers MC6821 or MC6822. Each chip has two 8 bit registers that can be programmed by software to be either an input or output. By "input" we mean that information can be brought into the microprocessor and by "output" we mean that information can be taken from the microprocessor. The MC6821 and MC6822 chips are called interface chips because they allow external devices to communicate with the microprocessor. A very useful feature of these chips is the fact that each bit of either of the 8 bit registers can be programmed to act as an input or an output. This gives complete flexibility.

The six bits for the D/A converter are located in a register that can be accessed by addressing memory location 65312 or \$FF20. The sound enable is in another register whose address is 65314 or \$FF22.

We want to do some experimenting with sounds which we can generate from the keyboard. To generate a sound we must do the following:

1. Enable the sound enable. This can be accomplished by POKING 255 into 65314.
2. Poke 255 into 65312 and then Poke 0 into 65312.
3. For continuous sound keep repeating step 2.


#### BASIC SOUNDS

Can sounds be generated by Basic by poking values into memory locations? We are not interested in the Basic "SOUND"

command because we can't get a feeling for what happens. Our only use will be to demonstrate how it works. What we want to do is to generate our own sounds by writing directly to the computer's 6 bit D/A converter. If you are not familiar with the computer's sounds then you might want to experiment with the sound command. The format is as follows:

```
SOUND 100,10
      or
SOUND T,D
```

From the keyboard type "SOUND 100,10" enter. You will notice a sound in your TV speaker. Change either of the numbers and the sound will change. There are two things that we can change which are the first number which represents the time, and the second number which represents the duration.



Digital Square wave  
Figure 1

### SQUARE WAVES

Last month we showed how a sine wave looked and this month we want to look and experiment with square waves. The sounds that we can produce when we alternately apply all ones and all zeros to the D/A converter is a square wave. As stated last month the sine wave is the basic waveform for all sounds. Due to a process called "FOURIER ANALYSIS" all linear waveforms can be reduced to a summation of

sine waves. Since we will be generating square waves it would be interesting to know how they relate to sine waves. A square wave can be produced by adding sinewaves of odd multiples of the frequency of the square wave. Another way to state this is to say that a square wave is composed of many sinewaves. For example if we have a square wave at 100 hertz then it will have sinewave components at 300 hertz, 500 hertz, 700 hertz, 900 hertz, etc,. This is a study in itself and we will not pursue it here.

We wrote a sound basic program to demonstrate how to generate sounds by writing to the D/A converter. This follows in the next section. There is a lot that can be done and we will continue next month.

### SOUND LEARNING PROGRAM

To demonstrate the principles covered in the preceding section we wrote a Basic program. We had two objectives in writing the program. The first was to demonstrate the maximum frequency that could be generated by alternately applying a "0" and then a "1" to the D/A converter using Basic. The second was to write a machine language subroutine that could be called from Basic to generate faster sounds similar to the Basic Sound commands.

### ML SUBROUTINE

First let's look at the Machine Language Subroutine. We reserved the following memories to be used by Basic and the ML subroutine.

- 500 - Used for sound duration
- 501 - Used for time (T)
- 510 - Delay counter for ML sub.

Refer back to Figure 1. The

time (T) is the time that the value applied to the D/A remains before changing. The duration (D) is actually a count of the number of cycles that are allowed.

The ML subroutine is carried with the basic program as data. Basic reads the bytes and stores the ML program in memory starting at 100 bytes above the end of the Basic Program. This makes it position independent. Let's look at the ML program. All numbers are decimal numbers.

```

10 LDB E 500 'Put the value in
    location 500 into the B
    register.
13 STB E 510 'Store B in 510
16 LDA E 501 'Load A with the
    value in 501
19 CLRB 'Clear B. Make all
    bits =0
20 STB E 65312 'Store B in
    65312 which writes to the
    D/A converter
23 DECA 'Decrement A or A=A-1
24 BNE 23 'Go to 23 if A is
    not 0
26 LDA E 500
29 LDB I 255 'Put the value 255
    in the B register.
31 STB E 65312 'Store B in
    65312
34 DECA ' A=A-1
35 BNE 34 'Go to 34 if A is
    not 0
37 DEC E 510 'Reduce the value
    in location 510 by 1.
40 TST E 510 'Is location
    510 =0?
43 BNE 16' Branch to 16 if 510
    is not a zero.
45 RTS 'Return from Sub.

```

The numbers are relative numbers since this is a position independent program. The first thing we do is transfer the duration time in location 500 to location 510. This is 10 and 13. Next we load "A" with the time in 500 and place a "0" to the input of the D/A by storing a 0 in 65312. This is 16-20. Next we go through a time delay

by decrementing A until it equals "0". This is 23 & 24. In 26 we load "A" with the time, in 27 we load "B" with 255 and in 31 we store "B" in 65312 putting all "1"s to the D/A converter. An equivalent time delay is in 34 & 35. We then decrement 510 which is the delay. If this is not zero then we go to 16 and send out another cycle.

Notice that if the time is shorter, then we can generate the waveform faster. This gives us higher frequencies. A longer time gives lower frequencies.

## SOUND DEMONSTRATION PROGRAM

To demonstrate the generation of sounds by writing directly to the D/A Converter we wrote the following program. The first part gives some random sounds to show some sound effects using the Basic Sound command. This is in statement 30.

### ML SUBROUTINE

Because Basic is too slow for most sounds we had to write a machine language subroutine which we can call from Basic. This is discussed elsewhere in this newsletter. We carried the machine language subroutine as data within the Basic program. Statement 80 is a delay to slow down the display. Statements 100-120 perform the transferring of the ML subroutine. We decided to place the ML subroutine 100 bytes above the end of the Basic program. The end of Basic vector is in location 27 & 28. Statements 100 and 110 calculate the beginning location for the ML subroutine. Statement 120 actually does the transferring.

Maximum Basic Rate

To show the maximum frequency of sound you can obtain using Basic for the memory pokes we wrote section 130-180. Statement 140 enables the sound output by poking a 255 into location 65315. In statements 150-170 we alternately POKE a "0" and then a "255" into location 65312. We do this for 200 times with the FOR - NEXT Loop. This will give a low frequency buzzing sound on your speaker.

### High Frequency Sound

Statements 200 - 270 are used with the ML Subroutine to generate higher frequency sounds than can be obtained with Basic. In 210 we go to locations 500 and 501 for the duration and time. These are printed in 220. The computer is forced to wait for a keyboard command in 230. If a "D" is pressed then you can enter a new duration. If a "T" is pressed then you can enter a new time. For any other key the sound is made with the duration and time displayed. You can continue with the previous sound by hitting any key besides a "T" or "D". The sound subroutine is called from statement 260. We used "Z" to represent the beginning of the ML Subroutine.

For different sounds change the value of "T". For a longer time you get a lower frequency, and higher frequencies are generated with lower times.

The duration is the length of the sounds.

### SOUND PROGRAM LISTING

NOTE: If a statement continues to the next line, we indented the second line for clarity. You do not need to enter the extra spaces when you type in the program.

```

10 CLS
20 PRINT"      SOUND LEARNING
   PROGRAM
25 PRINT"      THESE ARE RANDOM
   SOUNDS
30 FOR K=1 TO 100: X=RND (255):
   SOUND X,1:NEXTK
40 PRINT"      BY BILL
   CHAPPLE
50 PRINT"      COPYRIGHT (c)
   1984
60 PRINT"      DYNAMIC eLECTRONICS
   INC.
70 PRINT"THE NAME OF THIS PGM
   IS SND-1
80 PRINT:FORJ=1TO2000:NEXT
90 PRINT"TRANSFERING THE ML
   SUBROUTINE TO MEMORY.
100 Z=256*PEEK(27)+PEEK(28)+100
110 M=Z-1
120 FOR J=1 TO 36:READ A:V=M+J:
   POKE V,A: NEXT J
130 PRINT"THIS DEMONSTRATES THE
   MAXIMUM RATE THAT CAN BE
   OBTAINED FROM BASIC.
140 POKE65315,255
150 FOR X=1 TO 200
160 POKE 65312,0:POKE65312,255
170 NEXTX
180 PRINT"THIS IS THE END OF
   THIS PART":PRINT:PRINT
190 FOR K=1TO 2000:NEXT
200 PRINT"THIS IS THE MENUE.
   THESE ARE THE TIMES AND
   DURATIONS. IF YOU WANTTO
   CHANGE THEM ENTER 'T' OR
   'D' PRESS ENTER TO REPEAT
   THE LAST SOUND. MAX
   VALUE=255
210 D=PEEK(500):T=PEEK(501)
220 PRINT"DURATION="D,"TIME="T
230 A$=INKEY$: IFA$=""THEN230
240 IF A$="D" THEN INPUT
   "DURATION";D:POKE500,D
250 IF A$="T" THEN INPUT
   "TIME";T:POKE501,T
260 POKE65315,255:EXECZ
270 GO TO 210
280 DATA 246,1,244,247,1,
   254,182,1,245,95,247,
   255,32,74,38,253
290 DATA 182,1,245,198,255,
   247,255,32,74,38,253,
   122,1,254,125,1
300 DATA 254,38,227,57

```



**MERRY CHRISTMAS**  
**and a**  
**HAPPY NEW YEAR**

**OPERATING HINT**

**REDUCE DISPLAY NOISE**

Sometimes you can reduce noise on you television by folding the cable from your computer to the television's antenna terminal back on itself. Put rubber bands around the cord to hold it together. You can adjust the position of the rubber bands for minimum noise in the television. By folding the antenna cord back on itself you cancel stray noise that the cable might be picking up. Don't coil it up but keep it straight. Sometimes moving it to different positions helps.

\*\*\*\*\*

**SUBSCRIPTION  
DRIVE**

We have around 100 paid subscribers and will apply for a bulk rate the first of 1985. We will be required to mail a minimum of 200 pieces. We distribute around 100 as samples. If you will help us increase our number of subscribers we will give you \$3 for each new subscriber you get for us. This will not apply to your subscription but to each additional one. We are printing sales flyers for the newsletters which you can distribute to your friends. You can put your name on them so you can receive your \$3 when they subscribe. Write or call for a supply of flyers if you are interested.

**CLASSIFIED ADS**

Beginning with the February 1985 issue we will be taking classified and display ads. If you have something to sell you can advertise here. Also we will be reviewing Products and have a New Products Announcement section.

**CLASSIFIED AD RATES**

1. 10 cents a word.
2. Name, Address, & Telephone listed free.
3. Minimum ad cost \$3.00
4. Send payment with ad.
5. Closing date 10th of the preceeding month. Feb. ad closing is Jan 10.

**DISPLAY ADS**

1. Closing 10th of preceeding month.
2. Rates as follows:

Pages	1 time	2 times	3 times
1	\$25	\$22	20
1/2	13	15	13
1/4	10	8	7

**NEXT MONTH**

Make your own sound waveforms. Editorial on passing information between banks.

\*\*\*\*\*  
 \* Please sign me up for one year for the DYNAMIC COLOR NEWS SERVICE. I \*  
 \* Want to receive instruction on programming, Computer Theory, Operat- \*  
 \* ing Techniques, Computer Expansion plus the Individual Reply to my \*  
 \* Computer problems for a special of \$10 each. Also I understand that \*  
 \* there will be no charge for letters printed with answers in the \*  
 \* Newsletter. Cost \$15 USA & Canada, \$30 foreign. \*  
 \* \*  
 \* Name \_\_\_\_\_ Mail payment to \*  
 \* Address \_\_\_\_\_ Dynamic Electronics Inc \*  
 \* City \_\_\_\_\_ P. O. Box 896 \*  
 \* State & Zip \_\_\_\_\_ Hartselle, AL 35640 \*  
 \* Enclosed is a check \_\_\_ \*  
 \* charge to VISA \_\_\_ MC \_\_\_ Number \_\_\_\_\_ Exp. \_\_\_\_\_ \*  
 \* \*  
 \*\*\*\*\*

**DYNAMIC ELECTRONICS INC.**  
 P. O. Box 896 (205) 773-2758  
 Hartselle, AL 35640